

# ISBN Buchnummern in wissenschaftlicher Arbeit überprüfen

V. Thormählen, Okt. 2021

## Inhaltsverzeichnis

1	Aufgabenstellung .....	2
2	WORD-Dokumente .....	2
3	Prozeduren im Quelldokument .....	2
4	Praktisches Beispiel .....	2
5	Quellcode .....	3

## Abbildungen

Abbildung 1: Erfolgsmeldung .....	2
-----------------------------------	---

## Listings

Listing 1: Zieldokument auswählen .....	3
Listing 2 : Zieldokument im Dialog mit dem Benutzer auswählen .....	3
Listing 3: 17-stellige ISBN im Zieldokument finden .....	4
Listing 4: Prüfziffer berechnen .....	4

## Tabellen

Tabelle 1: VBA-Prozeduren im Quelldokument .....	2
--	---

## 1 Aufgabenstellung

In diesem Beitrag wird beschrieben, wie die ISBN<sup>1</sup> Buchnummern in einen Zieldokument mit dem VBA-Code in einem Quelldokument auf formale Richtigkeit überprüft werden können.

## 2 WORD-Dokumente

Unter *Quelldokument* wird hier ein WORD<sup>2</sup>-Dokument verstanden, das den VBA<sup>3</sup>-Code zur Überprüfung der ISBNs in einem Zieldokument beinhaltet. Dieses ist ebenfalls ein WORD-Dokument, das vom Benutzer im Dialog ausgewählt werden kann, um die ISBN-Buchnummern in einem dazugehörigen Literatur- und/oder Quellenverzeichnis auf formale Richtigkeit zu überprüfen.

## 3 Prozeduren im Quelldokument

Die im Quelldokument enthaltenen VBA-Prozeduren sind aus Tabelle 1 ersichtlich:

Name	Typ	Aufgabe
GetTgtFile	Subprozedur	Zieldokument auswählen
fctShowFileDialogue	Funktionsprozedur	Zieldokument im Dialog mit dem Benutzer auswählen.
FindISBN17	Subprozedur	17-stellige ISBN Buchnummern in Zieldokument finden
GetPZ	Subprozedur	Prüfziffer berechnen f. gefundene ISBN im Zieldokument

Tabelle 1: VBA-Prozeduren im Quelldokument

## 4 Praktisches Beispiel

Angenommen, im Zieldokument befindet sich folgende 17-stellige ISBN-Nummer:

*Hirschmann, Claudine ; Wanner, Inge: Wissenschaftliches Schreiben und Publizieren - Erläuterung für Studierende und Doktoranden. 1. Aufl. Norderstedt: BoD – Books on Demand, 2011. ISBN 978-3-842-33567-7. S. 1-176*

Die 17-stellige ISBN-Nummer (s. oben) beinhaltet 13 Datenstellen und 4 Gliederungszeichen. Nachdem die 17-stellige ISBN im Zieldokument gefunden wurde (vgl. Listing 3, Codezeile 12) erfolgt die entsprechende Prüfziffernberechnung mit der dafür geeigneten Funktionsprozedur GetPZ. Die genannten Gliederungszeichen werden vorher entfernt.

Stimmt die errechnete Prüfziffer mit der letzten Datenstelle der gefundenen ISBN-17 Buchnummer überein, ist die formelle Richtigkeit bestätigt (vgl. Listing 3, Codezeilen 26 und 27) und das Ergebnis wird wie folgt angezeigt:




Abbildung 1: Erfolgsmeldung bei Übereinstimmung der gefundenen Prüfziffer mit der errechneten

Bei Nichtübereinstimmung wird eine entsprechende Fehlermeldung ausgegeben.

<sup>1</sup> ISBN steht für International Standard Book Number

<sup>2</sup> WORD ist der Name eines verbreiteten Textverarbeitungsprogramms der Firma Microsoft.

<sup>3</sup> VBA steht für Visual Basic for Applications, einer Skriptsprache für die Office-Suite der Firma Microsoft.

## 5 Quellcode

```
1 Private Sub GetTgtFile()  
2 ' Aufgabe: Word-Dokument (=> Zieldokument) auswählen.  
3 Dim docTgt As Document ' Zieldokument  
4 Dim strFileNm As String ' Dateiname des Zieldokuments  
5 On Error GoTo Error_Point ' Fehlerbehandlung  
6 strFileNm = fctShowFileDialogue ' Zieldokument im Dialog auswählen  
7 If Len(strFileNm) > 0 Then  
8 ' Zieldokument öffnen  
9 Set docTgt = Documents.Open( _  
10 FileName:=strFileNm, _  
11 ReadOnly:=True, _  
12 AddToRecentFiles:=False, _  
13 Visible:=True)  
14 With docTgt  
15 ' Zieldokument aktivieren.  
16 .Activate  
17 ' ISBN im Zieldokument finden.  
18 Call FindISBN17(docTgt)  
19 End With  
20 Else  
21 MsgBox "Fehler bei der Auswahl der Zieldatei", vbCritical, "siehe GetTgtFile"  
22 End If  
23 Exit_Point:  
24 Exit Sub  
25 Error_Point:  
26 MsgBox "Laufzeitfehler: " & Err.Description, vbCritical, "GetTgtFile"  
27 Resume Exit_Point  
28 End Sub
```

Listing 1: Zieldokument auswählen

```
1 Function fctShowFileDialogue()  
2 ' Aufgabe: Zieldokument im Dialog mit dem Benutzer auswählen.  
3 With Application.FileDialog(msoFileDialogFilePicker)  
4 .AllowMultiSelect = False  
5 .Title = "Word-Datei auswählen"  
6 .InitialFileName = Options.DefaultFilePath(wdDocumentsPath) & "\*.docx"  
7 .ButtonName = "Auswählen"  
8 With .Filters  
9 .Clear  
10 .Add "Word-Dateien", "*.docx; *.docm", 1  
11 .Add "Alle Dateien", "*.*"  
12 End With  
13 .Show  
14 If .SelectedItems.Count = 0 Then  
15 fctShowFileDialogue = vbNullString  
16 Else  
17 fctShowFileDialogue = .SelectedItems(1)  
18 End If  
19 End With  
20 End Function
```

Listing 2: Zieldokument im Dialog mit dem Benutzer auswählen

```

1 Sub FindISBN17(ByRef docTgt As Document)
2     ' Aufgabe: 17-stellige ISBN in Word-Dokument finden:
3     ' 13 numerische Datenstellen + 4 Gliederungszeichen = 17 Stellen
4     ' Ruft die Funktionsprozedur GetPZ (Prüfziffernberechnung) auf.
5     Dim strISBN As String ' Internationale Standard Buchnummer
6     Dim strPZ As String * 1 ' Prüfziffer
7     Dim objRng As Range ' Textbereich
8     Dim intHits As Integer ' Trefferzähler
9     Set objRng = docTgt.Content
10    With objRng.Find
11        .ClearFormatting
12        .Text = "[0-9-]{17}" ' 13 Datenstellen + 4 Gliederungszeichen
13        .Replacement.Text = ""
14        .Forward = True
15        .Wrap = wdFindStop
16        .Format = False
17        .MatchCase = False
18        .MatchWholeWord = False
19        .MatchWildcards = True
20        .MatchSoundsLike = False
21        .MatchAllWordForms = False
22        .Execute
23        Do While .Found = True
24            strISBN = objRng.Text
25            strPZ = GetPZ(strISBN) ' Errechnete Prüfziffer
26            If CStr(Right(strISBN, 1)) = strPZ Then
27                MsgBox "PZ = " & CStr(strPZ) & " ist richtig.", vbExclamation,
28                "Prüfziffernberechnung für ISBN-17 = " & strISBN
29            Else
30                MsgBox "PZ = " & CStr(strPZ) & " ist falsch.", vbExclamation,
31                "Prüfziffernberechnung für ISBN-17 = " & strISBN
32            End If
33            .Execute
34        Loop
35    End With
36 End Sub

```

Listing 3: 17-stellige ISBN im Zieldokument finden

```

1 Private Function GetPZ(strISBN As String) As String
2     ' Aufgabe: Prüfziffer berechnen
3     Dim intLoop As Integer ' Schleifenzähler
4     Dim intMulti As Integer ' Multiplikator (1 bzw. 3 im Wechsel)
5     Dim intResult As Integer ' Produkt (Datenstelle * Multiplikator)
6     Dim intSum As Integer ' Produktsumme
7     Dim strNumber As String ' ISBN, aber ohne Gliederungszeichen
8     GetPZ = vbNullString
9     strNumber = Replace(strISBN, "-", "")
10    strNumber = Replace(strNumber, " ", "")
11    If Len(strNumber) = 13 Then
12        intSum = 0
13        intMulti = 1
14        For intLoop = 1 To Len(strNumber) - 1
15            intResult = Mid(strNumber, intLoop, 1) * intMulti
16            intSum = intSum + intResult
17            If intMulti = 1 Then intMulti = 3 Else intMulti = 1
18        Next intLoop
19        GetPZ = 10 - intSum Mod 10
20    Else
21        MsgBox "Fehler in Funktion GetPZ", vbCritical
22    End If
23 End Function

```

Listing 4: Prüfziffer berechnen